#

pkgname=xfdesktop
pkgver=4.8.2
pkgrel=1
pkgdesc="A desktop manager for Xfce"
arch=('ppc')
license=('GPL2')
url="http://www.xfce.org/"
groups=('xfce4')
depends=('dbus-glib' 'libwnck' 'libnotify' 'glib2' 'gtk2' \
				'desktop-file-utils' 'hicolor-icon-theme' \
				'garcon' 'xfconf' 'libxfce4util' 'libxfce4ui' 'exo>=0.6.0' \
				'thunar>=1.2.0')
makedepends=('intltool' 'pkg-config' 'gettext')
options=('!libtool')
install=${pkgname}.install
source=(http://archive.xfce.org/src/xfce/${pkgname}/4.8/${pkgname}-${pkgver}.tar.bz2
				src_xfce-desktop-menu.patch)
md5sums=('ed25d59f478afda552d121e96657d16f'
				'cf8fcc299a0d3129302675f9ea96e451')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"

	# Patch taken from FreshPorts
	patch -Np0 -i "${srcdir}"/src_xfce-desktop-menu.patch

  ./configure --prefix=/usr \
	--sysconfdir=/etc \
	--libexecdir=/usr/lib/xfce4 \
	--localstatedir=/var \
	--disable-static \
	--enable-nls \
	--enable-gio-unix \
	--enable-thunarx \
	--enable-exo \
	--enable-notifications
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
}
